# $Id$
# Maintainer: Jan de Groot <jgc@archlinux.org>
# Contributor: Arjan <arjan@archlinux.org>
# contributor Sarah Hay <sarahhay@mb.sympatico.ca>

pkgname=libdvdnav
pkgver=6.0.0
pkgrel=1
pkgdesc="The library for xine-dvdnav plugin."
arch=('x86_64')
license=('GPL')
url="https://www.videolan.org/developers/libdvdnav.html"
depends=('libdvdread')
makedepends=('git')
_commit=dcb9109e45ccd304ec82a7c7bf46cca63620adf9  # tags/6.0.0^0
source=("git+https://code.videolan.org/videolan/libdvdnav.git#commit=$_commit")
sha256sums=('SKIP')

pkgver() {
  cd $pkgname
  git describe --tags | sed -e 's/-/+/g'
}

prepare() {
  cd $pkgname
  autoreconf -fi
}

build() {
  cd $pkgname
  ./configure --prefix=/usr
  make
}

package() {
  cd $pkgname
  make DESTDIR="${pkgdir}" install
}
